ECE3120: Computer Systems
Chapter 7: Interfacing I/O Devices

Jeremy Langston

Today

- Interfacing with LCD

Liquid Crystal Display (LCD) (1 of 2)

- The basic construction of an LCD is illustrated in Figure 7.34.
- The most common type of LCD allows the light to pass through when activated.
- An LCD segment is activated when a low frequency bipolar signal in the range of 30 Hz to 1KHz is applied to it.
- LCD can display characters and graphics.

LCDs are often sold in a module with LCDs and controller unit built in.
The Hitachi HD44780 is the most popular LCD controller being used today.



Figure 7.34 A liquid crystal display (LCD)

A HD44780-Based LCD Kit (1 of 3)

- Display capability: 4 x 20
- Uses the HD44780 as the controller as shown in Figure 7.35.
- Pins DB7~DB0 are used to exchange data with the CPU.
- E input should be connected to one of the address decoder output or I/O pin.
- The RS signal selects instruction register (0) or data register (1).
- The VEE signal allows the user to adjust the LCD contrast.
- The HD44780 can be configured to display 1-line, 2-line, and 4-line information.
- The pin assignment for character-based LCD module with less than and more than 80 characters are shown in Table 7.7 and 7.8.



Figure 7.35 Block diagram of a HD44780-based LCD kit

A HD44780-Based LCD Kit (2 of 3)

Table 7.7 Pin assignment for displays with less than 80 characters

Pin No.	symbol	I/O	Function
1	VSS	-	Power supply (GND)
2	VCC	-	Power supply (+5V)
3	VEE	-	Contrast adjust
4	RS	I	0 = instruction input, 1 = data input
5	R/$\overline{W}$	I	0 = write to LCD, 1 = read from LCD
6	E	I	enable signal
7	DB0	I/O	data bus line 0
8	DB1	I/O	data bus line 1
9	DB2	I/O	data bus line 2
10	DB3	I/O	data bus line 3
11	DB4	I/O	data bus line 4
12	DB5	I/O	data bus line 5
13	DB6	I/O	data bus line 6
14	DB7	I/O	data bus line 7

A HD44780-Based LCD Kit (3 of 3)

Table 7.8 Pin assignment for displays with more than 80 characters

Pin No.	symbol	I/O	Function
1	DB7	I/O	data bus line 7
2	DB6	I/O	data bus line 6
3	DB5	I/O	data bus line 5
4	DB4	I/O	data bus line 4
5	DB3	I/O	data bus line 3
6	DB2	I/O	data bus line 2
7	DB1	I/O	data bus line 1
8	DB0	I/O	data bus line 0
9	E1	I	enable signal row 0 & 1
10	R/$\overline{W}$	I	0 = write to LCD, 1 = read from LCD
11	RS	I	0 = instruction input, 1 = data input
12	VEE	-	Contrastadjust
13	VSS	-	Powersupply(GND)
14	VCC	-	Powersupply(+5V)
15	E2	I	Enable signal row 2 & 3
16	N.C	-	

HD44780 Commands (1 of 4)

Table 7.9 HD44780U instruction set

Instruction	Code										Description	Execution time
	RS	R/$\overline{W}$	B7	B6	B5	B4	B3	B2	B1	B0		
Clear display	0	0	0	0	0	0	0	0	0	1	Clears display and returns cursor to the home position (address 0).	1.64 ms
Cursor home	0	0	0	0	0	0	0	0	1	*	Returns cursor to home position (address 0). Also returns display being shifted to the original position. DDRAM contents remain unchanged.	1.64 ms
Entry mode set	0	0	0	0	0	0	0	1	I/D	S	Set cursor move direction (I/D), specifies to shift the display (S). These operations are performed during data read/write.	40 µs
Display on/off control	0	0	0	0	0	0	1	D	C	B	Sets on/off of all display (D), cursor on/ off (C) and blink of cursor position character (B).	40 µs
Cursor /display shift	0	0	0	0	0	1	S/C	R/L	*	*	Sets cursor-move or display-(S/C), shift direction (R/L). DDRAM contents remains unchanged.	40 µs
Function set	0	0	0	0	1	DL	N	F	*	*	Sets interface data length (DL), number of display line (N) and character font (F).	40 µs
Set CGRAM address	0	0	0	1	CGRAM address						Sets the CGRAM address. CGRAM data is sent and received after this setting.	40 µs
Set DDRAM address	0	0	1	DDRAM address							Sets the DDRAM address. DDRAM data is sent and received after this setting.	40 µs
Read busy flag and address counter	0	1	BF	CGRAM/DDRAM address							Reads busy flag (BF) indicating internal operation is being performed and reads CGRAM or DDRAM address counter contents (depending on previous instruction).	0 µs
Write CGRAM or DDRAM	1	0	write data								Writes data to CGRAM or DDRAM.	40 µs
Read from CGRAM or DDRAM	1	1	read data								Reads data from CGRAM or DDRAM.	40 µs

HD44780 Commands (2 of 4)

Table 7.10 LCD instruction bit names

Bit name	Settings	
I/D	0 = decrement cursor position.	1 = increment cursor position
S	0 = no display shift.	1 = display shift
D	0 = display off	1 = display on
C	0 = cursor off	1 = cursor on
B	0 = cursor blink off	1 = cursor blink on
S/C	0 = move cursor	1 = shift display
R/L	0 = shift left	1 = shift right
DL	0 = 4-bit interface	1 = 8-bit interface
N	0 = 1/8 or 1/11 duty (1 line)	1 = 1/16 duty (2 lines)
F	0 = 5x8 dots	1 = 5 x 10 dots
BF	0 = can accept instruction	1 = internal operation in progress

HD44780 Commands (3 of 4)

- The HD44780 has a display data RAM (DDRAM) to store data to be displayed on the LCD.
- The address range of DDRAM for 1-line, 2-line, and 4-line LCDs are shown in Table 7.11a, 7.11b, and 7.11c.
- The HD44780 has a character generator ROM that can generates 5 × 8 or 5 × 10 character patterns from a 8-bit code.
- The user can rewrite character patterns into the character generator RAM (CGRAM).
- Up to eight 5 × 8 patterns or four 5 × 10 patterns can be programmed.

Table 7.11a DDRAM address usage for a 1-line LCD

Display size	Visible	
	character positions	DDRAMaddresses
1 * 8	00..07	0x00..0x07
1 * 16	00..15	0x00..0x0F
1 * 20	00..19	0x00..0x13
1 * 24	00..23	0x00..0x17
1 * 32	00..31	0x00..0x1F
1 * 40	00..39	0x00..0x27

Registers of HD44780

- The HD44780 has two 8-bit user accessible registers: instruction register (IR) and data register (DR).
- To write data into display data RAM or character generator RAM, the MCU writes into the DR register.
- The address of the data RAM should be set up with a previous instruction.
- The DR register is also used for data storage when reading data from DDRAM or CGRAM.
- The register selection is shown in Table 7.12.
- The HD44780 has a busy flag that is output from the DB7 pin.
- The HD44780 uses a 7-bit address counter to keep track of the address of the next DDRAM or CGRAM location to be accessed.

Table 7.12 Register selection

RS	R/$\overline{W}$	Operation
0	0	IR write as an internal operation (display clear, etc)
0	1	Read busy flag (DB7) and address counter (DB0 to DB6)
1	0	DR write as an internal operation (DR to DDRAM or CGRAM)
1	1	DR read as an internal operation (DDRAM or CGRAM to DR)

HD44780 Instructions (1 of 3)

- Clear display
  - Writes 0x20 (space character) to all DDRAM locations
  - Sets 0 to the address counter (return cursor to upper left corner of the LCD)
  - Sets increment mode
- Return home
  - Sets address counter to 0
  - DDRAM contents not changed
- Entry mode set
  - Sets incrementing or decrementing of the DDRAM address
  - Controls the shifting (shifts if S bit = 1) of the display
- Display on/off control
  - Turns on/off display
  - Turns on/off cursor
  - Turns on/off cursor blinking

HD44780 Instructions (2 of 3)

- Cursor or display shift
  - This function shifts the cursor position to the right or left without writing or reading display data.
  - The shifting is controlled by two bits as shown in Table 7.13.

Table 7.13 LCD Shift function

S/C	R/L	Operation
0	0	Shifts the cursor position to the left. (AC is decremented by 1)
0	1	Shifts the cursor position to the right. (AC is incremented by 1)
1	0	Shifts the entire display to the left. The cursor follows the display shift.
1	1	Shifts the entire display to the right. The cursor follows the display shift.

- Function set
  - Sets the interface length (DL bit) to be 4- or 8-bit
  - Selects the number of lines (N bit) to be one or two lines
  - Selects character font (F bit) to be 5 × 8 or 5 × 10

HD44780 Instructions (3 of 3)

- Set CGRAM address
  - This command contains the address to be written into the address counter.
- Set DDRAM address
  - This command allows the user to set the starting address to display information.
- Read busy flag and address
  - This command reads the busy flag and the address counter.
  - User can use this command to determine the LCD controller is ready to accept another command.
  - User can use this command to control where to start displaying information.

Interfacing the HD44780 with the HCS12

- One can treat the LCD kit as an I/O device and use an I/O port and several other I/O pins as control signals.
- The interface can be 4 bits or 8 bits.
- To read or write the LCD successfully, one must satisfy the timing requirements of the LCD. The timing diagrams for read and write are shown in Figure 7.37 and 7.38.



Figure 7.36a LCD interface example (8-bit bus, used in SSE256)



Figure 7.36b LCD interface example (4-bit bus, used in Dragon12)

- Procedure to send a command to the IR register
  - Step 1
    - Pull the RS and the E signals to low.
  - Step 2
    - Pull the R/W signal to low.
  - Step 3
    - Pull the E signal to high.
  - Step 4
    - Output data to the output port attached to the LCD data bus. One needs to configure the I/O Port for output before writing data to the LCD kit.
  - Step 5
    - Pull the E signal to low and make sure that the internal operation is complete.

- The procedure for writing a byte to the LCD data register
  - Step 1
    - Pull the RS signal to high.
  - Step 2
    - Pull the R/W signal to low.
  - Step 3
    - Pull the E signal to high.
  - Step 4
    - Output data to the I/O port attached to the LCD data bus.
  - Step 5
    - Pull the E signal to low and make sure that the internal operation is complete.
- These procedures need to be repeated once for an LCD kit with 4-bit interface.

- Write a function to send a command to the LCD kit
  - Most LCD commands are completed in 40 ms.
  - If the function waits for 40 ms after performing the specified operation, then most commands will be completed when the function returns.
  - The assembly code for the 8-bit interface is as follows:

```
lcdPort     equ   PTH                 ; LCD data port
lcdCtl      equ   PTK                 ; LCD control port
lcdE        equ   $80                 ; E signal pin (PK7)
lcdRW       equ   $20                 ; R/W signal pin (PK5)
lcdRS       equ   $10                 ; RS signal pin (PK4)
; the command is contained in A
cmd2lcd     bclr  lcdCtl,lcdRS+lcdRW        ; select instruction register and Write
            bset  lcdCtl,lcdE         ; pull the E signal high
            staa  lcdPort             ; send the command, along with RS, E signals
            nop
            nop
            bclr  lcdCtl,lcdE         ; pull the E signal low
            bset  lcdCtl,lcdRW        ; pull R/W to high
            ldy   #1                  ; adding this delay will complete the internal
            jsr   delayby50us         ; operation for most instructions
            rts
```

- The function to configure LCD sends four commands to the LCD kit
  - Entry mode set
  - Display on/off
  - Function set
  - Clear display

```
lcdDIR     equ     DDRH
lcdCtlDIR  equ     DDRK
openlcd    movb    #$FF,lcdDIR       ; configure port H for output
           bset    lcdCtlDir,$B0     ; configure control pins for output
           ldy     #5                ; wait for LCD to complete internal
           jsr     delayby100ms      ; configuration
           ldaa    #$38              ; set 8-bit data, 2-line display, 5x8 font
           jsr     cmd2lcd           ;       "
           ldaa    #$0F              ; turn on display, cursor, and blinking
           jsr     cmd2lcd           ;       "
           ldaa    #$06              ; move cursor right (entry mode set instruction)
           jsr     cmd2lcd           ;       "
           ldaa    #$01              ; clear LCD screen and return to home position
           jsr     cmd2lcd           ;       "
           ldy     #2                ; wait until "clear display" command is complete
           jsr     delayby1ms        ;       "
           rts
```

- Function to output a character to the LCD
  - The character to be output is in accumulator A.

```
putc2lcd bset   lcdCtl,lcdRS    ; select LCD Data register
         bclr   lcdCtl,lcdRW    ; enable write to LCD
         bset   lcdCtl,lcdE     ; pull E to high
         staa   lcdPort         ; send data to LCD
         nop                    ; provide enough length to E signal
         nop                    ;       "
         bclr   lcdCtl,lcdE     ; pull the E signal low
         bset   lcdCtl,lcdRW    ; pull R/W high to complete the write cycle
         ldy    #1              ; wait until the write operation is
         jsr    delayby50us     ; complete
         rts
```

- Function to output a string terminated by a NULL character
  - The string to be output is pointed to by index register X.

```
puts2lcd    ldaa    1,x+          ; get one character from the string
            beq     done_puts     ; reach NULL character?
            jsr     putc2lcd
            bra     puts2lcd
done_puts   rts
```

- Example 7.7 Write an assembly program to test the previous four subroutines by displaying the following messages on two lines:

  hello world!

  I am ready!

```
#include   "hcs12.inc"
lcdPort    equ     PTH          ; LCD data pins (PH7~PH0)
lcdDIR     equ     DDRH         ; LCD data direction port
lcdCtl     equ     PTK          ; LCD control port
lcdCtlDir  equ     DDRK         ; LCD control port direction
lcdE       equ     $80          ; E signal pin
lcdRW      equ     $20          ; R/W signal pin
lcdRS      equ     $10          ; RS signal pin
           org     $1500
           lds     #$1500       ; set up stack pointer
           jsr     openlcd      ; initialize the LCD
           ldx     #msg1lcd
           jsr     puts2lcd
           ldaa    #$C0         ; move to the second row
           jsr     cmd2lcd      ;            "
           ldx     #msg2lcd
           jsr     puts2lcd
           swi
msg1lcd    fcc     "hello world!"
           dc.b    0
msg2lcd    fcc     "I am ready!"
           dc.b    0
#include   “c:\miniide\delay.asm"          ; include delay routines here
; include the previous four LCD functions
```